以下のようなXMLサンプルがあるとき、XSLでHTMLタグに変換したいと思っています。

```

    <root>
        <section>
            <container>
                aaa
                <box>
                    book
                </box>
                bbb
                <box>
                    pen
                </box>
                ccc
                <superscript>
                    3
                </superscript>
                ddd
            </container>
        </section>
    </root>
    
```

XSLで、以下のような結果を得ることは出来るでしょうか？「aaa」「bbb」「ccc3ddd」をpタグでラップし、「box」にはdivタグを、「superscript」にはspanタグを割り当てたいと考えています。よろしければ、アドバイスお願いします。

```

    <div>
        <p>aaa</p>
        <div>book</div>
        <p>bbb</p>
        <div>pen</div>
        <p>ccc<span>3</span>ddd</p>
    </div>
    
```
CC BY-SA 3.0
2015-09-09T02:45:22.037
16413
2015-09-12T05:52:07.953
12259
question
xml xsl
XMLのテキストノードを要素ノードでラップしたい（条件あり）
XSLは実際に使ったことが無いのですが、DOMのパース方法をメタに記述できるものだというざっくりした理解でお話させていてだくと、  
一般的にDOM操作でテキストノードをパースさせようとすると記述が非常に冗長になりますよ。  
例えばboxタグの内容、実際のテキストノードとしては[CR][TAB][TAB][TAB][TAB]book[CR]のようにテキストノードに入っています。ブラウザで扱うとbookだけが見えますが、それはブラウザのレンダリングの規則がそうなっているからでしかないのです。  
たとえばsectionタグ、厳密にパースするとcontainerの後ろに[CR][TAB]というテキストノードが入っていると報告されます。タグの終わりに限らず、ヒトが見やすいように入れた改行も全部CRがあるといって報告されます。  
こういったものの解釈は結局人が手で与えなければいけなくて、かなり大変ですよ。

できればデータ構造をみなおして意味のあるテキストノードにはタグを振った方がいいと思います。
CC BY-SA 3.0
2015-09-10T01:07:09.717
16455
2015-09-10T01:07:09.717
8136
16413
answer
sken2 さんの回答を読んでいて思いついたのですが、強引にやればできなくはない、といった所でしょうか…。

**sample.xslt**

```

    <?xml version="1.0" encoding="UTF-8"?>
    <xsl:stylesheet version="2.0"
      xmlns:xsl="http://www.w3.org/1999/XSL/Transform">
    
      <xsl:output method="html" encoding="UTF-8"/>
    
      <xsl:template match="/">
        <html>
        <body>
        <xsl:apply-templates/>
        </body>
        </html>
      </xsl:template>
    
      <xsl:template match="section">
        <xsl:for-each select="container">
          <xsl:variable name="item" select="tokenize(normalize-space(.), ' ')"/>
          <div>
          <xsl:for-each select="text()|*">
            <xsl:choose>
              <xsl:when test="name(.)='box'">
                <xsl:variable name="n" select="position()-1"/>
                <p><xsl:value-of select="$item[$n]"/></p>
                <div><xsl:value-of select="normalize-space(.)"/></div>
              </xsl:when>
              <xsl:when test="name(.)='superscript'">
                <xsl:variable name="n" select="position()-1"/>
                <xsl:variable name="m" select="position()+1"/>
                <p>
                  <xsl:value-of select="$item[$n]"/>
                  <span><xsl:value-of select="normalize-space(.)"/></span>
                  <xsl:value-of select="$item[$m]"/>
                </p>
              </xsl:when>
            </xsl:choose>
          </xsl:for-each>
          </div>
        </xsl:for-each>
      </xsl:template>
    
    </xsl:stylesheet>
    
```

XSLT プロセッサとして [saxonb-xslt](http://saxon.sourceforge.net/) を使用しています。`tokenize`
関数を利用していますので、XSLT 2.0 対応のプロセッサが必要です。

`for-each` 〜 `choose` としているのはノードの順序保持のためです。ただ、やはり sken2
さんの言うとおり、データ構造を変更する方が良いかと思います。例えば、`aaa` や `ccc` を属性(attribute)にすればパースは簡単になります。

```

    <box id="aaa">book</box>
    <superscript id1="ccc" id2="ddd">3</superscript>
    
```
CC BY-SA 3.0
2015-09-10T03:16:50.627
16459
2015-09-10T03:16:50.627
16413
answer
今までいろいろXSLTで変換をやっていますが、入力XMLでもう少しちゃんとタグ付けをしていてくれたら良いのにと思ってもなかなかお客様の元データを直すわけにはいかず、XSLTスタイルイート側で吸収しなければならないことが多々ありました．

今回のご質問も"aaa","bbb","ccc"~"ddd"がpでタグ付けされていれば電車道ですね．でもそう行かないのでしょう．このような場合グルーピングの問題として解決することが出来るのではないか？と思います．つまりcontainer要素以下のnode()をbox要素かそうでないものかによってグルーピングするのです．

  * box要素の場合は個々にdiv要素に変換します．
  * 他のノードの場合はp要素にグルーピングして変換します

argusさんのスタイルシートを参考にさせていただき、XSLT 2.0のxsl:for-each-groupで実現したのが以下のコードです．

```

    <?xml version="1.0" encoding="UTF-8"?>
    <xsl:stylesheet xmlns:xsl="http://www.w3.org/1999/XSL/Transform"
        xmlns:xs="http://www.w3.org/2001/XMLSchema"
        exclude-result-prefixes="xs"
        version="2.0">
        <xsl:output method="html" encoding="UTF-8"/>
    
        <xsl:template match="/">
            <html>
                <body>
                    <xsl:apply-templates/>
                </body>
            </html>
        </xsl:template>
    
        <xsl:template match="section">
            <xsl:for-each select="container">
                <div>
                    <xsl:for-each-group select="node()" group-adjacent="name(.) eq 'box'">
                        <xsl:choose>
                            <xsl:when test="current-group()[self::box]">
                                <xsl:apply-templates select="current-group()"/>
                            </xsl:when>
                            <xsl:otherwise>
                                <p>
                                    <xsl:apply-templates select="current-group()"/>
                                </p>
                            </xsl:otherwise>
                        </xsl:choose>
                    </xsl:for-each-group>
                </div>
            </xsl:for-each>
        </xsl:template>
    
        <xsl:template match="box">
            <div>
                <xsl:apply-templates/>
            </div>
        </xsl:template>
    
        <xsl:template match="superscript">
            <span>
                <xsl:apply-templates/>
            </span>
        </xsl:template>
    
        <xsl:template match="text()">
            <xsl:value-of select="normalize-space(.)"/>
        </xsl:template>
    
    </xsl:stylesheet>
    
```

結果は以下のようになります．

```

    <html>
       <body>
          <div>
             <p>aaa</p>
             <div>book</div>
             <p>bbb</p>
             <div>pen</div>
             <p>ccc<span>3</span>ddd
             </p>
          </div>
       </body>
    </html>
    
```

それなりに自然な流れのスタイルシートになっていると思います．（何故か"ddd"のあとで改行されていてそれだけ疑問ですが？？？）

参考にしていただければ幸いです．
CC BY-SA 3.0
2015-09-12T05:43:58.727
16541
2015-09-12T05:52:07.953
2015-09-12T05:52:07.953
9503
9503
16413
answer